/*=========================================================================
Elite College as Engines of Upward Mobility: Evidence from Colombia's Ser Pilo Paga
Authors: Juliana Londoño-Vélez, Catherine Rodriguez, Fabio Sánchez
and Luis Esteban Álvarez-Arango

Creation date: June 6th, 2025
--------------------------------------------------------------------------
Input for Table 6
=========================================================================*/
    

/*
P6020 sexo
P6030S1 mes nacimiento
P6030S2 año nacimiento
P6040 edad
P6170 ¿actualmente ... Asiste a la escuela, colegio o universidad?
P6210 ¿cuál es el nivel educativo más alto alcanzado por ... Y el último año o grado aprobado en este nivel?
P6210S1 Grado
ESC escolaridad
P1806 En caso de que le ofrecieran un empleo a ... ¿Cuál sería la remuneración o el salario mensual más bajo por el que aceptaría?
DSCY Desempleo coyuntural
DSI Desocupados
P7430 ¿... Ha trabajado alguna vez por lo menos durante dos semanas consecutivas?
P7440 ¿cuánto hace que ... Trabajó por última vez?
P1807 En caso de que le ofrecieran un empleo a ... ¿Cuál sería el salario o remuneración mensual más baja por la que aceptaría?
INI Inactivos
P388 ¿en qué departamento realiza … principalmente este trabajo?
P6440 ¿para realizar este trabajo, tiene usted algún tipo de contrato?
P6450 ¿el contrato es verbal o escrito?
P6460 ¿el contrato de trabajo de ... Es a término indefinido o a término fijo?
P6426 ¿cuanto tiempo lleva ... Trabajando en esta empresa, negocio, industria, oficina, firma o finca de manera continua?
P6500 Antes de descuentos ¿cuánto ganó ... El mes pasado en este empleo?
P6765 En la semana pasada, ¿cuál de las siguientes formas de trabajo realizó:
P6750 ¿cuál fue la ganancia neta o los honorarios netos de ... Esa actividad, negocio, profesión o finca, el mes pasado?
P6760 ¿a cuántos meses corresponde lo que recibió?
OCI Población ocupada
P6430 En este trabajo ... Es:
INGLABO Ingresos laborales
P6920 ¿está ... Cotizando actualmente a un fondo de pensiones?
*/

local meses Enero Febrero Marzo Abril Mayo Junio Julio Agosto Septiembre Octubre Noviembre Diciembre
foreach x of local meses{
	use "${raw}/GEIH/`x'/Cabecera - Características generales (Personas)", clear
	keep DIRECTORIO SECUENCIA_P ORDEN P6020 P6030S1 P6030S3 P6040 P6170 P6210 P6210S1 ESC fex_c_2011 AREA MES P6220
	merge 1:1 DIRECTORIO SECUENCIA_P ORDEN using "${raw}/GEIH/`x'/Cabecera - Desocupados", keepusing(DSI DSCY P1806) nogen
	merge 1:1 DIRECTORIO SECUENCIA_P ORDEN using "${raw}/GEIH/`x'/Cabecera - Inactivos", keepusing(P7430 P7440 P1807 INI) nogen
	merge 1:1 DIRECTORIO SECUENCIA_P ORDEN using "${raw}/GEIH/`x'/Cabecera - Ocupados", keepusing(P388 P6440 P6450 P6460 P6426 P6500 P6765 P6750 P6760 OCI P6430 INGLABO P6920) nogen
	tempfile `x'
	save ``x''
}

clear
foreach x of local meses{
	append using ``x''
}

keep if P6040>17 & P6040<61 // edad
keep if P6210>4 // nivel educativo
drop if P6210==9
keep if P6220>1 & P6220<6
drop if INI==1
preserve
	import excel "${raw}/1.1.INF_Serie histórica.xlsx", clear sheet("Hoja1") firstrow case(lower)
	keep if año==2019
	drop año
	tempfile ipc
	save `ipc'
restore

destring MES, replace
rename MES mes
merge m:1 mes using `ipc' // diciembre 2021
replace INGLABO=INGLABO/índice*100
replace P6500=P6500/índice*100
recode INGLABO P6500 (.=0)
drop if P6920==3 // pensionados
gen informal=1 if P6920==2
replace informal=0 if P6920==1
gen ingreso_formal=INGLABO if informal==0
gen ingreso_informal=INGLABO if informal==1
gen desocupado=1 if DSI==1
replace desocupado=0 if desocupado==.
collapse INGLABO ingreso_formal ingreso_informal informal desocupado [pw=fex_c_2011], by(P6040)
replace INGLABO=round(INGLABO)
export excel "${Intermediate}/salario_edad.xlsx", replace firstrow(variables)